

This is an illustration of the use of the SVD to identify the pseudoinverse associated with problem 3.3.6. In this case, the computation of an exact pseudoinverse by this method is much more complicated than the use of the normal equations, which are available since the columns of A are linearly independent. The SVD gives so much more information about A in this case that it gets in the way of finding the pseudoinverse. This detracts from neither its role in providing a theoretical foundation or its effectiveness as part of a numerical method.

Initial setup

```
[ > with(LinearAlgebra):
```

This makes the new (as of Maple7) LinearAlgebra package available. It includes the shortcuts used below to enter the original matrix and construct the auxiliary matrices appearing in the exact computation of its SVD.

```
[ > A:=<<1|1>,<2|-1>,<-2|4>>;
```

$$A := \begin{bmatrix} 1 & 1 \\ 2 & -1 \\ -2 & 4 \end{bmatrix}$$

```
[ > AT:=Transpose(A);
```

$$AT := \begin{bmatrix} 1 & 2 & -2 \\ 1 & -1 & 4 \end{bmatrix}$$

```
[ > ATA:=AT.A;
```

$$ATA := \begin{bmatrix} 9 & -9 \\ -9 & 18 \end{bmatrix}$$

```
[ > (eA,vA):=Eigenvectors(ATA);
```

$$eA, vA := \begin{bmatrix} \frac{27}{2} + \frac{9}{2}\sqrt{5} \\ \frac{27}{2} - \frac{9}{2}\sqrt{5} \end{bmatrix}, \begin{bmatrix} 1 & 1 \\ -\frac{1}{2} - \frac{1}{2}\sqrt{5} & -\frac{1}{2} + \frac{1}{2}\sqrt{5} \end{bmatrix}$$

The "Eigenvectors" function in this package produces a vector of eigenvalues and a matrix whose columns are the eigenvectors associated with those values in the same order. This format was introduced with the LinearAlgebra package. The parallel assignment to a sequence of names was not present in earlier versions, but it appears to have been introduced quietly. The help page for the Eigenvectors function was the first place I saw it.

Step by step construction of SVD.

First, extract the eigenvectors and find their lengths.

```
> vA1:=vA[1..2,1];
```

$$vA1 := \begin{bmatrix} 1 \\ -\frac{1}{2} - \frac{1}{2}\sqrt{5} \end{bmatrix}$$

```
> vA2:=vA[1..2,2];
```

$$vA2 := \begin{bmatrix} 1 \\ -\frac{1}{2} + \frac{1}{2}\sqrt{5} \end{bmatrix}$$

```
> l1:=simplify(sqrt(vA1.vA1));
```

$$l1 := \frac{1}{2}\sqrt{10+2\sqrt{5}}$$

```
> l2:=simplify(sqrt(vA2.vA2));
```

$$l2 := \frac{1}{2}\sqrt{10-2\sqrt{5}}$$

Now, build the orthogonal matrix.

```
> Q2:=(1/l1)*vA1|(1/l2)*vA2>;
```

$$Q2 := \begin{bmatrix} \frac{1}{2\sqrt{10+2\sqrt{5}}} & \frac{1}{2\sqrt{10-2\sqrt{5}}} \\ -\frac{1}{2} - \frac{1}{2}\sqrt{5} & -\frac{1}{2} + \frac{1}{2}\sqrt{5} \\ \frac{1}{2\sqrt{10+2\sqrt{5}}} & \frac{1}{2\sqrt{10-2\sqrt{5}}} \end{bmatrix}$$

Check of orthogonality.

Checking orthogonality with these exact quantities involves some effort.

> `Transpose(Q2).Q2;`

$$\left[4 \frac{1}{10+2\sqrt{5}} + \frac{4 \left(-\frac{1}{2} - \frac{1}{2}\sqrt{5} \right)^2}{10+2\sqrt{5}}, \right.$$

$$\left. 4 \frac{1}{\sqrt{10+2\sqrt{5}} \sqrt{10-2\sqrt{5}}} + \frac{4 \left(-\frac{1}{2} - \frac{1}{2}\sqrt{5} \right) \left(-\frac{1}{2} + \frac{1}{2}\sqrt{5} \right)}{\sqrt{10+2\sqrt{5}} \sqrt{10-2\sqrt{5}}} \right]$$

$$\left[4 \frac{1}{\sqrt{10+2\sqrt{5}} \sqrt{10-2\sqrt{5}}} + \frac{4 \left(-\frac{1}{2} - \frac{1}{2}\sqrt{5} \right) \left(-\frac{1}{2} + \frac{1}{2}\sqrt{5} \right)}{\sqrt{10+2\sqrt{5}} \sqrt{10-2\sqrt{5}}}, \right.$$

$$\left. 4 \frac{1}{10-2\sqrt{5}} + \frac{4 \left(-\frac{1}{2} + \frac{1}{2}\sqrt{5} \right)^2}{10-2\sqrt{5}} \right]$$

> `simplify(%);`

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

[The eigenvalues we have are the squares of the singular values.

> `sVA:=simplify(map(sqrt,eA));`

$$sVA := \begin{bmatrix} \frac{3}{2}\sqrt{5} + \frac{3}{2} \\ \frac{3}{2}\sqrt{5} - \frac{3}{2} \end{bmatrix}$$

> `Sigma:=Matrix(1..3,1..2,sVA,shape=diagonal);`

$$\Sigma := \begin{bmatrix} \frac{3}{2}\sqrt{5} + \frac{3}{2} & 0 \\ 0 & \frac{3}{2}\sqrt{5} - \frac{3}{2} \\ 0 & 0 \end{bmatrix}$$

These values have now been inserted along the diagonal of a matrix of the same shape as A. Multiplication gives a matrix whose columns are orthogonal of lengths given by the singular values.

> `AQ2:=simplify(A.Q2);`

$$AQ2 := \begin{bmatrix} -\frac{\sqrt{5}-1}{\sqrt{10+2\sqrt{5}}} & \frac{\sqrt{5}+1}{\sqrt{10-2\sqrt{5}}} \\ \frac{5+\sqrt{5}}{\sqrt{10+2\sqrt{5}}} & -\frac{-5+\sqrt{5}}{\sqrt{10-2\sqrt{5}}} \\ -4\frac{2+\sqrt{5}}{\sqrt{10+2\sqrt{5}}} & 4\frac{-2+\sqrt{5}}{\sqrt{10-2\sqrt{5}}} \end{bmatrix}$$

Another check

The form used here is not easily seen to be the same as the one originally used to express the eigenvalues. Here are some techniques for reformatting to recognize equality.

```
> simplify(AQ2[1..3,1].AQ2[1..3,1]);
```

$$18 \frac{5+2\sqrt{5}}{5+\sqrt{5}}$$

```
> rationalize(%);
```

$$-\frac{9}{10} (5+2\sqrt{5})(-5+\sqrt{5})$$

```
> expand(%);
```

$$\frac{27}{2} + \frac{9}{2}\sqrt{5}$$

```
> eA[1];
```

$$\frac{27}{2} + \frac{9}{2}\sqrt{5}$$

A similar computation for the second column can be skipped.

[

[Now to find columns of Q1.

> `cQ1:=(1/Sigma[1,1])*AQ2[1..3,1];`

$$cQ1 := \begin{bmatrix} \frac{\sqrt{5}-1}{\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)\sqrt{10+2\sqrt{5}}} \\ \frac{5+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)\sqrt{10+2\sqrt{5}}} \\ -4\frac{2+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)\sqrt{10+2\sqrt{5}}} \end{bmatrix}$$

> `cQ2:=(1/Sigma[2,2])*AQ2[1..3,2];`

$$cQ2 := \begin{bmatrix} \frac{\sqrt{5}+1}{\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)\sqrt{10-2\sqrt{5}}} \\ \frac{-5+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)\sqrt{10-2\sqrt{5}}} \\ 4\frac{-2+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)\sqrt{10-2\sqrt{5}}} \end{bmatrix}$$

A third column is needed. All we know is that it must be orthogonal to those we have. A general method would be to use Householder matrices to synthesize an orthogonal matrix containing the columns we know. Since we are in 3 dimensions, we have a shortcut -- the cross product -- that finds a vector orthogonal to two given vectors. We know that it will be a unit vector.

> `cQ3:=simplify(CrossProduct(cQ1,cQ2));`

$$cQ3 := \begin{bmatrix} \frac{2}{3} \\ \frac{-2}{3} \\ \frac{-1}{3} \end{bmatrix}$$

[We have the columns, so we build the matrix.

> Q1 := <cQ1 | cQ2 | cQ3>;

$$Q1 := \begin{bmatrix} -\frac{\sqrt{5}-1}{\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)\sqrt{10+2\sqrt{5}}} & \frac{\sqrt{5}+1}{\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)\sqrt{10-2\sqrt{5}}} & \frac{2}{3} \\ \frac{5+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)\sqrt{10+2\sqrt{5}}} & -\frac{-5+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)\sqrt{10-2\sqrt{5}}} & -\frac{2}{3} \\ -4\frac{2+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)\sqrt{10+2\sqrt{5}}} & 4\frac{-2+\sqrt{5}}{\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)\sqrt{10-2\sqrt{5}}} & -\frac{1}{3} \end{bmatrix}$$

Finding the pseudoinverse from its SVD.

The nonzero eigenvalues need to be inverted and arranged in a matrix of the appropriate shape.

> `InvSV:=map(x->1/x,sVA);`

$$\text{InvSV} := \begin{bmatrix} \frac{1}{\frac{3}{2}\sqrt{5} + \frac{3}{2}} \\ \frac{1}{\frac{3}{2}\sqrt{5} - \frac{3}{2}} \end{bmatrix}$$

> `SigmaPlus:=Matrix(1..2,1..3,InvSV,shape=diagonal);`

$$\text{SigmaPlus} := \begin{bmatrix} \frac{1}{\frac{3}{2}\sqrt{5} + \frac{3}{2}} & 0 & 0 \\ 0 & \frac{1}{\frac{3}{2}\sqrt{5} - \frac{3}{2}} & 0 \end{bmatrix}$$

| [Now we multiply the factors in the SVD of the pseudoinverse...

> `APlus:=Q2.SigmaPlus.Transpose(Q1);`

`APlus:=`

$$\left[\begin{array}{l} -2 \frac{\sqrt{5}-1}{(10+2\sqrt{5})\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)^2} + \frac{2(\sqrt{5}+1)}{(10-2\sqrt{5})\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)^2}, \\ 2 \frac{5+\sqrt{5}}{(10+2\sqrt{5})\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)^2} - \frac{2(-5+\sqrt{5})}{(10-2\sqrt{5})\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)^2}, \\ -8 \frac{2+\sqrt{5}}{(10+2\sqrt{5})\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)^2} + \frac{8(-2+\sqrt{5})}{(10-2\sqrt{5})\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)^2} \end{array} \right]$$

$$\left[\begin{array}{l} -2 \frac{\left(-\frac{1}{2}-\frac{1}{2}\sqrt{5}\right)(\sqrt{5}-1)}{(10+2\sqrt{5})\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)^2} + \frac{2\left(-\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)(\sqrt{5}+1)}{(10-2\sqrt{5})\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)^2}, \\ 2 \frac{\left(-\frac{1}{2}-\frac{1}{2}\sqrt{5}\right)(5+\sqrt{5})}{(10+2\sqrt{5})\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)^2} - \frac{2\left(-\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)(-5+\sqrt{5})}{(10-2\sqrt{5})\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)^2}, \\ -8 \frac{\left(-\frac{1}{2}-\frac{1}{2}\sqrt{5}\right)(2+\sqrt{5})}{(10+2\sqrt{5})\left(\frac{3}{2}\sqrt{5}+\frac{3}{2}\right)^2} + \frac{8\left(-\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)(-2+\sqrt{5})}{(10-2\sqrt{5})\left(\frac{3}{2}\sqrt{5}-\frac{3}{2}\right)^2} \end{array} \right]$$

... and try to put it in a reasonable form.

> `simplify(%);`

$$\left[\begin{array}{ccc} \frac{16}{3} \frac{1}{(\sqrt{5}-1)^2 (\sqrt{5}+1)^2} & \frac{16}{3} \frac{1}{(\sqrt{5}-1)^2 (\sqrt{5}+1)^2} & 0 \\ & \frac{2}{9} & \frac{1}{9} \\ & & \frac{2}{9} \end{array} \right]$$

Trying to get a simple form.

Maple can be stupid! We can see that a simpler form is possible, but nothing seems to work.

```
> map(expand,%);
```

$$\begin{bmatrix} \frac{16}{3} \frac{1}{(\sqrt{5}-1)^2 (\sqrt{5}+1)^2} & \frac{16}{3} \frac{1}{(\sqrt{5}-1)^2 (\sqrt{5}+1)^2} & 0 \\ \frac{2}{9} & \frac{1}{9} & \frac{2}{9} \end{bmatrix}$$

```
> simplify(%,sqrt);
```

$$\begin{bmatrix} \frac{16}{3} \frac{1}{(\sqrt{5}-1)^2 (\sqrt{5}+1)^2} & \frac{16}{3} \frac{1}{(\sqrt{5}-1)^2 (\sqrt{5}+1)^2} & 0 \\ \frac{2}{9} & \frac{1}{9} & \frac{2}{9} \end{bmatrix}$$

Finally, the right approach.

```
> map(rationalize,%);
```

$$\begin{bmatrix} \frac{1}{3} & \frac{1}{3} & 0 \\ \frac{2}{9} & \frac{1}{9} & \frac{2}{9} \end{bmatrix}$$

```
> APlus:=%;
```

$$APlus := \begin{bmatrix} \frac{1}{3} & \frac{1}{3} & 0 \\ \frac{2}{9} & \frac{1}{9} & \frac{2}{9} \end{bmatrix}$$

This is the pseudoinverse of A.



Solving exercise 3.3.6

```
> b:=<1,2,7>;
```

$$b := \begin{bmatrix} 1 \\ 2 \\ 7 \end{bmatrix}$$

```
> APlus.b;
```

$$\begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

These are the coordinates with respect to the columns of A. We now find the components of b in the column space and its orthogonal complement.

```
> p:=A.%;
```

$$p := \begin{bmatrix} 3 \\ 0 \\ 6 \end{bmatrix}$$

```
> q=b-p;
```

$$q = \begin{bmatrix} -2 \\ 2 \\ 1 \end{bmatrix}$$

Since p is just the projection into the column space of A, it could be found directly by row operations. The appearance of the inverse of ATA in the solution of the normal equations in the case where A has linearly independent columns is one way to describe the solution of these equations. It is not the only description. Its main virtue is that it gives an explicit formula showing the linearity of the function computing x from b.

```
> MatrixInverse(ATA).AT;
```

$$\begin{bmatrix} \frac{1}{3} & \frac{1}{3} & 0 \\ \frac{2}{9} & \frac{1}{9} & \frac{2}{9} \end{bmatrix}$$

Because the columns of A were linearly independent, the description of the solution of the normal equations gave the pseudoinverse using only simple operations.